
Subject: Calculating malaria prevalence (Tanzania: Standard AIS, 2011-12)
Posted by NinaB on Mon, 21 Nov 2016 11:19:54 GMT
View Forum Message <> Reply to Message

Hi,

I'm trying to reproduce the malaria prevalences (by mRDT and microscopy) for children 6-59
months which can be found in the Tanzania AIS 11/12 final report on page 157. I watched the
DHS youtube videos on sampling and weighting and read some other threads in this forum.
However, I was not able to get exactly the same numbers as in the report. I'm using the household
member dataset (PR) and my stata code so far looks like this: 

*Parasitemia (via microscopy) in children 6-59 months
                gen micmalpos=0
                replace micmalpos=(hml32==1) 
                lab var micmalpos "Parasitemia (via microscopy) in children 6-59 months"

*Parasitemia (via mRDT) in children 6-59 months
                gen rdtmalpos=0
                replace rdtmalpos=(hml35==1) 
                lab var rdtmalpos "Parasitemia (via mRDT) in children 6-59 months"
	
				
*Creating weight variable
gen wgt=hv005/1000000 

*Survey setting the data
svyset [pw=wgt], psu(hv021) strata(hv023)

*Proportion of children 6-59 months old with malaria infection (microscopy) by Region
svy: mean micmalpos if hc1>=6 & hc1<=59 & hml33==0, over(hv024)

*Proportion of children 6-59 months old with malaria infection (mRDT) by Region
svy: mean rdtmalpos if hc1>=6 & hc1<=59 & hml33==0, over(hv024)

Does anyone have an idea what I'm doing wrong. Thanks for helping me.
Nina

Subject: Re: Calculating malaria prevalence (Tanzania: Standard AIS, 2011-12)
Posted by Liz-DHS on Wed, 23 Nov 2016 21:11:15 GMT
View Forum Message <> Reply to Message

A response from malaria expert, Dr. Lia Florey,
Hi Nina,

It looks as though you are using the correct approach. When examining this particular dataset I
see that a crosstab of hml33 and hml32 produces some unusual results. Same with the crosstab
of hml33 and hml35. In order to replicate the estimates from the final report I would recommend
the following changes to your code:

*Proportion of children 6-59 months old with malaria infection (microscopy) by Region
svy: mean micmalpos if hc1>=6 & hc1<=59 & hv103==1 & (hml32==0|hml32==1), over(hv024)

*Proportion of children 6-59 months old with malaria infection (mRDT) by Region
svy: mean rdtmalpos if hc1>=6 & hc1<=59 & hv103==1 & (hml35==0|hml35==1), over(hv024)

The hv103 specifies that children who slept in the household the night before the survey (de
facto). 
You might also try running tabulations instead of means:
svy: tab hv024 micmalpos if hc1>=6 & hc1<=59 & hv103==1 & (hml32==0|hml32==1), row ci per
svy: tab hv024 rdtmalpos if hc1>=6 & hc1<=59 & hv103==1 & (hml35==0|hml35==1), row ci per

I hope that this solves your problem.

Best,
Lia

